A DIVERSITY INDEX APPROACH TO ANALYSIS OF STANDARDIZATION IN PREHISTORIC POTTERY

Barbara L. Stark
Department of Anthropology
Arizona State University

Joseph T. Hepworth
Department of Psychology
Arizona State University

Introduction

It is frequently assumed that occupational specialization in prehistoric pottery production will be accompanied by various changes in pottery assemblages. One possible change is toward greater standardization in particular forms, sizes, or decorations. If so, diversity indices offer one approach to recording and comparing degrees of standardization. Here we summarize some of the advantages and disadvantages of an index of widespread utility and present a computer program to calculate it. A brief illustration draws on pottery data from a Mesoamerican archaeological site. For our purposes we will ignore the rather complex set of issues surrounding the assumption that specialist pottery may display greater standardization in order to concentrate on assessment of the index and presentation of the program for it.

The Use of H'

The index we discuss is one of a number that have been used to compare communities in quantitative ecology (Peet 1974). The ecologists' concept of diversity or heterogeneity is not a simple one because often it comprises two distinct factors: the number of species in a "community" and the frequency of each. The former aspect of diversity of sometimes termed "richness" while the latter is "evenness" (Odum 1971:149). For a variety of mathematical and conceptual reasons, one of the most desirable indices is H', the Shannon-Wiener (Pielou 1974:290) or Shannon-Weaver (Poole 1974:391) index, originally developed as an information measure (Pielou 1977:298):

$$H' = \sum_{i=1}^{s} p_i \log p_i$$

where s is the number of species or categories and p may be estimated by the proportion of the total number of individuals or cases that pertain to the *i*th category. Thus the proportion, p_i, is $\frac{n_i}{N}$ where n_i is the frequency of cases in the _i_th category and N is the total number of cases. Logarithms in the H' formula may be calculated with various bases (Pielou 1974:290); we will follow common practice and use the base e.

Although H' is a statistic based on infinitely large populations (Pielou 1974:290), it can be used if a suitably large random sample is drawn. By large sample we refer to sample size not fraction. However, most archaeological collections (e.g., of rim sherds) are samples from a site acquired from areal collection units. The collection thus may randomly sample the area of the site, but it cannot, for example, directly sample the universe of rim sherds at the site. While one might hope that random areal sampling would provide a representative sample of the pottery, we are not in a position to assure this. However, it seems reasonable to attempt to work with areally drawn samples as a practical compromise.

Alternatives to H' are less satisfactory for the study of archaeological standardization (diversity). H, the Brillouin index (Pielou 1974:304), deals with finite, fully censused populations. It has no standard error, and any two different values of H are consequently significantly different (Poole 1974:389). Poole (1974:397) states that "the two measures H and H' are most strongly affected by the abundances of the middle species of a community rather than by the common or rare species." However, Peet (1974:296) stresses that H' is most sensitive to rarer categories.

Another index, the Simpson index, "expresses the dominance of or concentration of abundance into the one or two commonest species of the community" (Poole 1974:396), not the overall evenness of the abundances of the categories (or species). In the future the Simpson index and variants of it (Peet 1974:291) should be investigated more closely for applications to the study of pottery specialization, but, unlike H', a basis for comparison of two collections using the t test or another inferential statistic has not been derived and published to our knowledge.

In practice, application of the H' statistic to a sample too small to include representatives of all of the categories in the collection will distort the results by underrepresenting diversity since rarer categories are not included. Large sample sizes will mitigate this distortion.

The formulae for calculating values of H' and estimates that allow use of the t test are cumbersome (Hutcheson 1970). However, these calculations are vital to allow comparison of different collections of differing sample sizes to determine if they are likely to represent the same population in terms of diversity. A computer program is the only feasible means to assure accurate calculations for large samples and to permit comparisons of numerous collections or data sets. Appendix 1 provides the program developed by Hepworth.

An additional problem arises with multiple comparisons because these involve application of the t test to a series of two-sample comparisons. This increases the likelihood of Type I error (to falsely reject the hypothesis of no difference). While in many cases we might wish to compare diversities of a number of different kinds of pottery at once, in fact we are limited to pair-wise comparisons. Use of a stringent confidence interval (e.g., .01) is one means to reduce somewhat the probability of Type I error in multiple comparisons. Additionally, development of focused expectations and comparisons will reduce the complications of inductive "pattern search" using H'. For example, in our application we have targeted three categories of pottery of particular interest rather than addressing a larger set of comparisons.

Example Application

The H' diversity index may be applied to the measurement of the degree of standardization in pottery in several ways. Rice (1981) used type-variety groups, which were employed in the description of assemblages from different occupations of the Maya site of Barton Ramie (Gifford 1976). She could not apply the H' statistic to published decorative, technological, and form attributes (although they were analyzed in regard to their frequencies per 1000 sherds) because the latter were not tabulated originally in independent, mutually exclusive categories.

In our application, the ceramic typological approach and attribute observations have been tailored to provide information about standardization. Since we do not know what functional or other categories were relevant prehistorically to the makers and users of the vessels, we can only attempt to carefully note and control for a set of variables that may have been significant. Using these variables we form sets of pottery to compare in regard to standardization. Specifically, some consistency in the observable characteristics of clay and temper, of surface treatment and decoration, and of overall vessel form (e.g., bowls versus necked jars) defined sets of rim sherds. These sets were compared to each other.

In our example, selected for the sake of simplicity, the variable used for comparison was rim form. Control for clay, temper, and surface treatment was achieved through grouping the sherds into named types similar to those of the type-variety system. The named types were established by Shook and Hatch (1978). Control for overall vessel form was established by using sherds with enough of the sidewall preserved to allow categorization. Rim shapes were coded using 83 possible categories defined in a manual.

The assumption in our study is that more specialized producers manufacturing a particular ware will produce less variation in rim shape than will a presumably larger number of less specialized producers. Obviously we must also assume that the sherds in the three categories selected for analysis give us adequate representation of products in general use by households. It is possible, for example, that the sherds produced by a single family which was self-sufficient in pottery would not differ from an equal number of sherds that represent the handiwork of a single specialist potter working during a comparable number of production episodes. The basic idea, then, is to compare rim shape among selected kinds of pottery in a given assemblage to detect differences in diversity. Individual rim shape determinations are the cases, and the rim form categories constitute the counterpart of species in biological applications.

In our illustration we compare three categories of pottery from Late Preclassic (500-0 B.C.) deposits at the center of El Balsamo on the Pacific coast of Guatemala in the department of Escuintla (Shook and Hatch 1978). The data derive from the El Balsamo Residential Project, which tested habitational refuse and structures in a 60 by 80 m field area directly east of the center. A random sample approximating two percent of 1 by 1 m test pits was excavated. These data were supplemented by additional excavations to expose structures.

The three categories of pottery are Sacatepequez White Paste White Ware (SAWH), El Balsamo-Monte Alto Brown Ware (BROW), and Miscellaneous Orange Ware (MIOR). SAWH closely resembles white ware pottery found in the Guatemalan highlands during the Preclassic (Shook and Hatch 1978; Stark 1982). Sources of white firing clay have been identified in the Valley of Guatemala (Rice 1978) and appear to be quite restricted geographically. Consequently SAWH found at El Balsamo could have been acquired through exchange with highland groups. SAWH is found in low frequencies at El Balsamo, which might be expected in the case of non-local pottery.

In contrast, BROW and MIOR are both plentiful at El Balsamo and include thicker vessels which perhaps implies a wider variety of utilitarian domestic functions. BROW and MIOR, unlike SAWH and some other low-frequency pottery categories, do not seem to be represented at highland sites (Stark 1982). Therefore, BROW and MIOR are more likely to represent local wares produced on the coast, perhaps at El Balsamo and/or in the vicinity.

For the H' analysis, the incidence of different lip shapes among the three categories was compared. In the initial analysis we used only rim sherds which had a sufficient portion of the sidewall preserved to allow determination of probable vessel form. Bowls, necked jars, and neckless jars (tecomates) were represented. Bowl-like forms were further classified as convex, (i.e., restricted) or straight to outflaring forms. The expectation is that SAWH, as a possibly imported ware, could have been produced by part or full-time specialists in the highlands since it may have been produced in excess of local demand there. To analyze SAWH we were forced to combine both convex and outflaring bowl-like forms to obtain a better sample size, n = 23. Even so, sample size remains a problem. For MIOR and BROW, sample size is not particularly problematic, and we can examine convex and straight to outflaring bowl-like forms either separately or pooled (to create categories more strictly comparable to the combined bowl-like forms for SAWH). A sample of output comparing SAWH bowl-like vessels (SAWHBOWL) to a combination of MIOR convex and straight to outflaring vessels (MIOROFBL, MIORCOBL) in regard to the diversity of lip forms is:

```
DIVERSITY COMPARISON OF LIP SHAPES OF SACATEPEQUEZ WHITE
PASTE WHITE WARE TO MISCELLANEOUS ORANGE WARE AT EL BALSAMO,
ALL BOWL-LIKE VESSELS.

                TEST COMPARING DIVERSITIES FOR

                (1) SAWHBOWL  VS  (2) MIOROFBL
                                      MIORCOBL

H1 =     0.64285773    VARH1 =    0.04853729    N1 =    23.

H2 =     2.62987614    VARH2 =    0.01873225    N2 =   126.

T =        -7.6611     WITH       43.0094 DEGREES OF FREEDOM
```

At the .01 significance level, the above two H' values are significantly different using the t test. The same result occurs with a comparison of SAWH to pooled BROW bowls. Not surprisingly, BROW and MIOR pooled bowls are not significantly different from each other in terms of diversity. To examine the robustness of the contrast with SAWH, we also compared SAWH bowls separately to convex vessels of each of the local wares (n = 30 for MIOR, n = 82 for BROW) and to straight to outflaring vessels for each (n = 96 for MIOR, n = 276 for BROW). Only BROW convex bowl lips were not significantly different in diversity from SAWH bowl lips. Of course, these latter sets of BROW and MIOR categories are not strictly comparable to SAWH because more stringently defined groups of vessel forms were used for the "local" categories. Nevertheless, in most cases the pottery of probable local origin still was significantly more diverse.

Do these results indicate greater standardization for SAWH? This appears to be the case, but, unfortunately, the small sample of SAWH lips may fail to represent rarer lip forms, which makes a conclusion of greater standardization questionable. There is no direct solution to the sample size problem without further excavation to enlarge the collection, but we can attempt to circumvent it indirectly by adding SAWH rim sherds which did not have enough of the sidewall attached to allow categorization of vessel form (n = 26). Although undoubtedly these added sherds comprise some forms other than bowl-like vessels, it is probable that most of them derive from that kind of vessel because bowl-like forms comprise 62% of the determinable vessel forms (n = 37).

We performed the same set of comparisons described previously using SAWH bowl-like forms combined with those SAWH rim sherds for which vessel form could not be determined (total n = 49). The resulting statistical decisions were identical to the prior analyses with one exception. Both BROW and MIOR convex bowl-like vessels were not judged significantly different from the combined SAWH values (previously only BROW convex bowls produced this result when compared to SAWH).

It is noteworthy that the enlarged data set for SAWH was still judged significantly different from the combined (straight to outflaring and convex) BROW bowl-like vessels and from the similar set of MIOR sherds. These comparisons are particularly important because they use vessel form restrictions more comparable to the SAWH data set; hence, we give more weight to them. Our conclusion, then, is that SAWH does appear to be more standardized in lip forms than either MIOR or BROW when relatively comparable controls are exercised in defining sets of sherds. (Clearly one potentially could manipulate diversity results arbitrarily if one set of sherds is defined to allow less variability in vessel forms and another to allow more. Our partitioning of MIOR and BROW bowl-like vessels was simply to probe the robustness of the contrast with SAWH using more stringently defined groups of BROW and MIOR.)

Conclusion

A relatively rare and possibly imported category of El Balsamo pottery was shown to be more highly standardized in lip forms than other pottery which may have been produced in the coastal zone where the site is located. That the more standardized pottery may have been imported lends support to the inference that the potters who produced it were more specialized. However, we would stress that the controlled investigation of pottery standardization in relation to specialization is an area in which many questions remain. One pressing need is studies of standardization in relation to known contexts of production using ethnographic data sets. With other archaeological data it will be advisable to consider more than a single variable for comparison. We were forced to restrict our comparisons to the single variable of lip form even though we would have preferred to use orifice diameter as well.

The H' statistic appears to be a useful measure for analysis of pottery standardization, but there are other potential applications. One example is comparisons of artifact diversity in assemblages or subsets of them. For certain analytic purposes, limited activity or seasonal site assemblages could be usefully contrasted with permanent sites in terms of artifact diversity. Perhaps sites in central place hierarchies can be contrasted in regard to artifact diversity to clarify economic and social roles of centers.

Acknowledgments

We thank James Collins and Dennis Young at ASU for their statistical assistance. The analysis of prehistoric pottery using H' was begun under NSF grant BNS-791286 to analyze collections from Veracruz, Mexico. The approach was extended to El Balsamo, where archaeological fieldwork was supported by the Social Science Research Council with added funds from an Arizona State University Faculty Grant-in Aid and the College of Liberal Arts. We particularly thank Lic. Francis Polo Sifontes, Director of the IDEAH in Guatemala, for his interest in and support of the work at El Balsamo, and William Swezey and Christopher Lutz, Co-Directors of CIRMA, for their kind hospitality during laboratory analysis. C. William Clewlow suggested that the Residential Project conduct investigations at El Balsamo and encouraged our work, and we are very grateful for the permission and support of the site owner Sr. Julio Herrera.

Bibliography

Gifford, James C.
1976 Prehistoric pottery analysis and the ceramics of Barton Ramie in the Belize Valley. Memoirs of the Peabody Museum of Archaeology and Ethnology, Harvard University 18.

Hutcheson, Kermit
1970 A test for comparing diversities based on the Shannon formula. Journal of Theoretical Biology 29:151-154.

Odum, Eugene P.
1971 Fundamental of Ecology, third edition. W.B. Saunders Company. Philadelphia.

Peet, Robert K.
1974 The measurement of species diversity. Annual Review of Ecology and Systematics 5:285-307.

Pielou, E.C.
1974 Population and Community Ecology: Principles and Methods. Gordon and Breach, Science Publishers. New York.
1977 Mathematical Ecology. John Wiley & Sons. New York.

Poole, Robert W.
1974 An Introduction to Quantitative Ecology. McGraw-Hill Book Company. New York.

Rice, Prudence M.
1978 Ceramic continuity and change in the Valley of Guatemala: A technological analysis. In The Ceramics of Kaminaljuyu, ed. by R.K. Wetherington, pp. 401-510. The Pennsylvania State University Press. University Park.
1981 Evolution of specialized pottery production: A trial model. Current Anthropology 22:3:219-240.

Shook, Edwin, and Marion Hatch
1978 The ruins of El Balsamo, Department of Escuintla, Guatemala. Journal of New World Archaeology 3:1.

Stark, Barbara L.
1982 Pottery in relation to economy and social organization at El Balsamo, Escuintla, Guatemala. XVII Mesa Redonda of the Sociedad Mexicana de Antropologia, Chiapas, Mexico, August 1981. In press.

APPENDIX

by

Joseph T. Hepworth

Program Description

Purpose

This program computes estimates of H', a commonly used index of diversity, for two groups, their corresponding variance estimates, and their subsequent *t* statistic and degrees of freedom. The formulae used for these computations were taken from Hutcheson (1970). The formula used for computing the variances in this program is an approximation to the exact calculation. This approximation is more accurate as sample size increases.

Limitations

(1) No more than 9 "title" cards may be used to document each run.
(2) Up to 99 separate comparisons (t-tests) may be made in a single run.
(3) No more than 10 unique grouping names may be specified for inclusion in each group.

Preparation of the Data

The program expects cards in the following sequence:
(1) Control Card

card columns	contents
1	Total number of title cards which will immediately follow this control card (maximum 9, minimum 0). If a 0 is indicated in column 1, no user supplied title cards will be included.

2-3	Total number of comparisons (t tests) requested (maximum 99). If fewer than 10 comparisons are requested, this value should be right justified, i.e., punched in column 3.

(2) Title Cards

Immediately following the Control Card should be as many Title Cards as indicated in column 1 of the Control Card. If column 1 of the Control Card is 0, no user supplied Title Cards are to be included and the user should skip to the Format Card.

card columns	contents
1-80	Title to be printed on output (any alphanumeric characters may be used).

(3) Format Card

Two variables are read in at a time, the first being a group or category (species equivalent) name and the second being a cell count (frequency of occurrence) for that group.

card columns	contents
1-80	Variable format for one cell (must begin and end with parentheses and include a group code in A format, up to A8, followed by the cell count in F format). For example: (6X,A8,20X,F3.0).

(4) Comparison Specification Card

This card indicates the number of unique names (up to 10) which should be collectively considered as a single group. In other words, this card allows the user to pool sets of data to form a composite category.

card columns	contents
1-2	Total number of unique group names to be considered collectively as specifying Group 1 (maximum 10).
3-4	Total number of unique group names to be considered collectively as specifying Group 2 (maximum 10).

(5) Group 1 Card

In 8-column fields, this card will contain up to 10 names which collectively will be considered as Group 1. There should be as many 8-column fields with names in them as indicated in columns 1-2 of the Comparison Specification Card. (Note that the cases of a group need not be consecutive in the data file.)

card column	contents
1-8	Name of first Group 1 subgroup.
9-16	Name of second Group 1 subgroup, etc., until
73-80	Name of tenth Group 1 subgroup. The total number of names must equal the number indicated in columns 1-2 of the Comparison Specification Card. Unused subgroup fields should be left blank.

(6) Group 2 Card

In 8-column fields, this card will contain up to 10 names which collectively will be considered as Group 2. There should be as many 8-column fields with names in them as indicated in columns 3-4 of the Comparison Specification Card.

card columns	contents
1-8	Name of first Group 2 subgroup.
9-16	Name of second Group 2 subgroup, etc., until
73-80	Name of tenth Group 2 subgroup. The total number of names must equal the number indicated in columns 3-4 of the Comparison Specification Card. Unused subgroup fields should be left blank.

The Comparison Specification Card, along with its accompanying Group 1 Card and Group 2 Card, would then be repeated for a total of up to 99 times depending upon the number of comparisons indicated in columns 2-3 of the Control Card.

<u>Deck Set-up and Control Language for ASU Amdahl</u>

```
//    JOB
//STEP1  EXEC  FORTXCLG
//FORT.SYSIN  DD  *
         .
         .
         .
```

```
 fortran program
       .
       .
       .
//GO.SYSIN  DD  *
301
DIVERSITY COMPARISON OF LIP SHAPES OF SACATEPEQUEZ WHITE
PASTE WHITE WARE TO MISCELLANEOUS ORANGE WARE AT EL BALSAMO,
ALL BOWL-LIKE VESSELS.
(A8,6X,F4.0)
0102
SAWHBOWL
MIOROFBLMIORCOBL
//GO.FT08F001  DD  *
       .
       .
       .
     data
       .
       .
       .
//
```

Program

```
C*************************************************************************
C*
C*     A PROGRAM FOR COMPARING DIVERSITIES BASED
C*     ON THE SHANNON FORMULA
C*
C*     DERIVED FROM K. HUTCHESON, J. THEOR. BIOL. 1970
C*
C*     WRITTEN BY J. T. HEPWORTH, DEPT. OF PSYCHOLOGY, ASU 1981
C*
C*************************************************************************
      DOUBLE PRECISION DSET1(10),DSET2(10),DATA1(200,2),DATA2(200,2),
     1TITLE(9,20),DATA(5000,2),FMT(18)
C
C     READING IN THE CONTROL CARD FROM UNIT 5
C
      READ(5,340)ITITLE,ICOMP
C
C     CHECKING IF TITLE CARDS ARE TO FOLLOW
C     AND READING IN TITLE CARDS
C
      IF (ITITLE.EQ.0) GO TO 6
      DO 5 I=1,ITITLE
    5 READ (5,350) (TITLE(I,J) ,J=1,20)
C
C     READING IN THE INPUT FORMAT CARD
C
      READ(5,360) FMT
C
C     READING IN THE DATA
C
    6 DO 8 I=1,5000
      READ(8,FMT,END=10) (DATA(I,J),J=1,2)
    8 IOBSER=IOBSER + 1
C
C     STARTING DO LOOP TO BE EXECUTED ONCE FOR EACH COMPARISON
C
   10 DO 200 K=1,ICOMP
C
C     INITIALIZING VARIABLES TO 0
C
      D1N=0
      D2N=0
      H1=0
      H2=0
      ALPHA1=0
      ALPHA2=0
      BETA1=0
      BETA2=0
      GAMMA1=0
      GAMMA2=0
```

```
C
C     INITIALIZING COUNTS TO 1
C
      ID1C=1
      ID2C=1
C
C     READING IN THE CONTRAST CARD
C
      READ(5,390) I1SETS,I2SETS
C
C     READING DATA SET CARDS INDICATING WHICH TWO DATA SETS
C     ARE TO BE COMPARED
C
      READ(5,300) (DSET1(J),J=1,10)
      READ(5,300) (DSET2(J),J=1,10)
C
C     SELECTING THE DATA
C
      DO 40 I=1,IOBSER
        DO 14 J=1,I1SETS
   14    IF(DATA(I,1).EQ.DSET1(J))GO TO 20
        DO 17 J=1,I2SETS
   17    IF(DATA(I,1).EQ.DSET2(J))GO TO 30
        GO TO 40
   20   DO 25 M=1,2
   25   DATA1(ID1C,M)=DATA(I,M)
        ID1C=ID1C+1
        GO TO 40
   30   DO 35 M=1,2
   35   DATA2(ID2C,M)=DATA(I,M)
        ID2C=ID2C+1
   40 CONTINUE
C
C     CORRECTING CELL COUNTS
C
   45 ID1C=ID1C-1
      ID2C=ID2C-1

C
C     COMPUTING N FOR DATA SET 1
C
      DO 50 I=1,ID1C
   50   D1N=D1N + DATA1(I,2)
C
C     COMPUTING N FOR DATA SET 2
C
      DO 60 I=1,ID2C
   60   D2N=D2N + DATA2(I,2)
```

```
C
C     COMPUTING H FOR DATA SET 1
C
      DO 70 I=1,ID1C
   70 H1=H1+((DATA1(I,2)/D1N)*(DLOG(DATA1(I,2)/D1N)))
      H1=H1 * (-1.0)
C
C     COMPUTING H FOR DATA SET 2
C
      DO 80 I=1,ID2C
   80 H2=H2+((DATA2(I,2)/D2N)*(DLOG(DATA2(I,2)/D2N)))
      H2=H2 * (-1.0)
C
C     COMPUTING ALPHA, (NI/N)*(LN(NI/N))**2, FOR DATA SET 1
C
      DO 90 I=1,ID1C
   90 ALPHA1=ALPHA1+(DATA1(I,2)/D1N)*(DLOG(DATA1(I,2)/D1N))**2
C
C     COMPUTING ALPHA FOR DATA SET 2
C
      DO 100 I=1,ID2C
  100 ALPHA2=ALPHA2+(DATA2(I,2)/D2N)*(DLOG(DATA2(I,2)/D2N))**2
C
C     COMPUTING BETA,N/NI, FOR DATA SET 1
C
      DO 110 I=1,ID1C
  110 BETA1=BETA1+(D1N/(DATA1(I,2)))
C
C     COMPUTING BETA FOR DATA SET 2
C
      DO 120 I=1,ID2C
  120 BETA2=BETA2+(D2N/(DATA2(I,2)))
C
C     COMPUTING GAMMA, N/NI(LN(NI/N)), FOR DATA SET 1
C
      DO 130 I=1,ID1C
  130 GAMMA1=GAMMA1+(D1N/(DATA1(I,2)))*(DLOG(DATA1(I,2)/D1N))
C
C     COMPUTING GAMMA FOR DATA SET 2
C
      DO 140 I=1,ID2C
  140 GAMMA2=GAMMA2+(D2N/(DATA2(I,2)))*(DLOG(DATA2(I,2)/D2N))
C
C     COMPUTING VARH1
C
      VARH1=(ALPHA1-H1**2)/D1N + (ID1C-1)/(2*(D1N**2)) +
     1(-1.0 + BETA1-GAMMA1-(BETA1*H1))/(6*(D1N**3))
```

```
C
C     COMPUTING VARH2
C
      VARH2=(ALPHA2-H2**2)/D2N + (ID2C-1.0)/(2.0*(D2N**2)) +
     1(-1.0 + BETA2-GAMMA2-(BETA2*H2))/(6.0*(D2N**3))
C
C     COMPUTING DEGREES OF FREEDOM
C
      DF=((VARH1+VARH2)**2)/((VARH1**2/D1N)+(VARH2**2/D2N))
C
C     COMPUTING T
C
      T=(H1-H2)/SQRT(VARH1+VARH2)
C
C     WRITING TITLES
C
      WRITE(6,370)
      IF (ITITLE.EQ.0) GO TO 180
      DO 170 I=1,ITITLE
  170 WRITE(6,380) (TITLE(I,J),J=1,20)
  180 WRITE(6,330)
      WRITE(6,400)DSET1(1),DSET2(1)
      DO 190 I=2,10
  190 WRITE(6,410)DSET1(I),DSET2(I)
C
C     CHECKING FOR UNUSED SUBGROUPS
C
      DO 194 J=1,I1SETS
        DO 192 I=1,ID1C
  192   IF (DSET1(J).EQ.DATA1(I,1))GO TO 194
      WRITE(6,420)DSET1(J)
  194 CONTINUE
      DO 198 J=1,I2SETS
        DO 196 I=1,ID2C
  196   IF (DSET2(J).EQ.DATA2(I,1))GO TO 198
      WRITE(6,420)DSET2(J)
  198 CONTINUE
C
C     WRITING RESULTS
C
  200 WRITE(6,320)H1,VARH1,D1N,H2,VARH2,D2N,T,DF
      STOP
```

```
C
C     FORMATS
C
  300 FORMAT(10A8)
  320 FORMAT('0',5X,'H1 = ',F12.8,3X,'VARH1 = ',F12.8,3X,
     1'N1 = ',F5.0/'0',5X,'H2 = ',F12.8,3X,'VARH2 = ',
     2F12.8,3X,'N2 = ',F5.0/'0',5X,'T = ',F12.4,3X,
     3'WITH ',F12.4,' DEGREES OF FREEDOM')
  330 FORMAT('0'///16X,'TEST COMPARING DIVERSITIES FOR'/)
  340 FORMAT(I1,I2)
  350 FORMAT(20A4)
  360 FORMAT(18A4)
  370 FORMAT('1')
  380 FORMAT(1X,20A4)
  390 FORMAT(2I2)
  400 FORMAT(16X,'(1) ',A8,2X,'VS',2X,'(2) ',A8)
  410 FORMAT(20X,A8,10X,A8)
  420 FORMAT(3X,'*****WARNING:  SUBGROUP ',A8,' WAS NOT ENCOUNTERED ',
     1'IN DATA*****')
      END
```

Sample Data

SAWHBOWLA1	19
SAWHBOWLA2	2
SAWHBOWLF3	1
SAWHBOWLE1	1
MIOROFBLZ2	1
MIOROFBLU1	1
MIOROFBLR2	1
MIOROFBLQ5	2
MIOROFBLP5	1
MIOROFBLP2	1
MIOROFBLP1	1
MIOROFBLL4	1
MIOROFBLJ2	3
MIOROFBLJ1	3
MIOROFBLH1	1
MIOROFBLG2	3
MIOROFBLG1	2
MIOROFBLF3	1
MIOROFBLE1	8
MIOROFBLD3	1
MIOROFBLC3	2
MIOROFBLC2	1
MIOROFBLC1	3
MIOROFBLB4	1
MIOROFBLB2	2
MIOROFBLB1	5
MIOROFBLA2	5
MIOROFBLA1	46
MIORCOBLZ2	1
MIORCOBLE1	3
MIORCOBLD3	1
MIORCOBLB4	2
MIORCOBLB1	1
MIORCOBLA2	2
MIORCOBLA1	16
MIORCOBLM4	1
MIORCOBLK1	1
MIORCOBLF1	1
MIORCOBLB3	1